


# Hurkle

Hurkle? A Hurkle is a happy beast and lives in another galaxy on a planet named Lirht that has three moons. Hurkle are favorite pets of the Gwik, the dominant race of Lirht and . . . well, to find out more, read "The Hurkle is a Happy Beast," a story in the book *A Way Home* by Theodore Sturgeon.

In this program a shy hurkle is hiding on a 10 by 10 grid. Homebase is point 0,0 in the *Southwest* corner. Your guess as to the gridpoint where the hurkle is hiding should be a pair of whole numbers, separated by a comma. After each try, the computer will tell you the approximate direction to go look for the Hurkle. You get five guesses to find him; you may change this number in Line 110, although four guesses is actually enough.

This program was written by Bob Albrecht of People's Computer Company.



HURKLE  
CREATIVE COMPUTING MORRISTOWN, NEW JERSEY

A HURKLE IS HIDING ON A 10 BY 10 GRID. HOMEBASE ON THE GRID IS POINT 0,0 AND ANY GRIDPOINT IS A PAIR OF WHOLE NUMBERS SEPARATED BY A COMMA. TRY TO GUESS THE HURKLE'S GRIDPOINT. YOU GET 5 TRIES. AFTER EACH TRY, I WILL TELL YOU THE APPROXIMATE DIRECTION TO GO TO LOOK FOR THE HURKLE.

GUESS # 1 ? 5,5  
GO SOUTHEAST

GUESS # 2 ? 6,4  
GO SOUTH

GUESS # 3 ? 6,3  
GO SOUTH

GUESS # 4 ? 6,2

YOU FOUND HIM IN 4 GUESSES!

LET'S PLAY AGAIN. HURKLE IS HIDING.

GUESS # 1 ? 5,5  
GO NORTHWEST

GUESS # 2 ? 3,8  
GO NORTHWEST

GUESS # 3 ? 2,9

YOU FOUND HIM IN 3 GUESSES!

```

10 PRINT TAB(33);"HURKLE"
20 PRINT TAB(15);"CREATIVE COMPUTING MORRISTOWN, NEW JERSEY"
30 PRINT:PRINT:PRINT
110 N=5
120 G=10
210 PRINT
220 PRINT "A HURKLE IS HIDING ON A";G;"BY";G;"GRID. HOMEBASE"
230 PRINT "ON THE GRID IS POINT 0,0 AND ANY GRIDPOINT IS A"
240 PRINT "PAIR OF WHOLE NUMBERS SEPARATED BY A COMMA. TRY TO"
250 PRINT "GUESS THE HURKLE'S GRIDPOINT. YOU GET";N;"TRIES."
260 PRINT "AFTER EACH TRY, I WILL TELL YOU THE APPROXIMATE"
270 PRINT "DIRECTION TO GO TO LOOK FOR THE HURKLE."
280 PRINT
285 A=INT(G*RND(1))
286 B=INT(G*RND(1))
310 FOR K=1 TO N
320 PRINT "GUESS #";K;
330 INPUT X,Y
340 IF ABS(X-A)+ABS(Y-B)=0 THEN 500
350 REM PRINT INFO
360 GOSUB 610
370 PRINT
380 NEXT K
410 PRINT
420 PRINT "SORRY, THAT'S";N;"GUESSES."
430 PRINT "THE HURKLE IS AT ";A;"",B
440 PRINT
450 PRINT "LET'S PLAY AGAIN. HURKLE IS HIDING."
460 PRINT
470 GOTO 285
500 REM
510 PRINT
520 PRINT "YOU FOUND HIM IN";K;"GUESSES!"
540 GOTO 440
610 PRINT "GO ";
620 IF Y=B THEN 670
630 IF Y<B THEN 660
640 PRINT "SOUTH";
650 GO TO 670
660 PRINT "NORTH";
670 IF X=A THEN 720
680 IF X<A THEN 710
690 PRINT "WEST";
700 GO TO 720
710 PRINT "EAST";
720 PRINT
730 RETURN
999 END

```